

clonevar — Clone existing variable[Description](#)[Remarks and examples](#)[Quick start](#)[Acknowledgments](#)[Menu](#)[Also see](#)[Syntax](#)

Description

`clonevar` generates *newvar* as an exact copy of an existing variable, *varname*, with the same storage type, values, and display format as *varname*. *varname*'s variable label, value labels, notes, and characteristics will also be copied.

Quick start

Copy contents, label, and value label of `v1` to `newv1`

```
clonevar newv1 = v1
```

Copy observations from `v2` to `newv2` where `v2` is less than 30

```
clonevar newv2 = v2 if v2 < 30
```

Copy the first 20 observations of `v3` to `newv3`

```
clonevar newv3 = v3 in f/20
```

Same as above

```
clonevar newv3 = v3 in 1/20
```

Menu

Data > Create or change data > Other variable-creation commands > Clone existing variable

Syntax

```
clonevar newvar = varname [if] [in]
```

Remarks and examples

stata.com

`clonevar` has various possible uses. Programmers may desire that a temporary variable appear to the user exactly like an existing variable. Interactively, you might want a slightly modified copy of an original variable, so the natural starting point is a clone of the original.

▷ Example 1

We have a dataset containing information on modes of travel. These data contain a variable named `mode` that identifies each observation as a specific mode of travel: air, train, bus, or car.

```
. use https://www.stata-press.com/data/r18/travel
(Modes of travel)
```

```
. describe mode
```

Variable name	Storage type	Display format	Value label	Variable label
mode	byte	%8.0g	travel	Travel mode alternatives

```
. label list travel
```

```
travel:
```

```
 1 Air
 2 Train
 3 Bus
 4 Car
```

To create an identical variable identifying only observations that contain air or train, we could use `clonevar` with an `if` qualifier.

```
. clonevar airtrain = mode if mode == 1 | mode == 2
(420 missing values generated)
```

```
. describe mode airtrain
```

Variable name	Storage type	Display format	Value label	Variable label
mode	byte	%8.0g	travel	Travel mode alternatives
airtrain	byte	%8.0g	travel	Travel mode alternatives

```
. list mode airtrain in 1/5
```

	mode	airtrain
1.	Air	Air
2.	Train	Train
3.	Bus	.
4.	Car	.
5.	Air	Air

The new `airtrain` variable has the same storage type, display format, value label, and variable label as `mode`. If `mode` had any characteristics or notes attached to it, they would have been applied to the new `airtrain` variable, too. The only differences in the two variables are their names and values for bus and car.



□ Technical note

The `if` qualifier used with the `clonevar` command in example 1 referred to the values of `mode` as 1 and 2. Had we wanted to refer to the values by their associated value labels, we could have typed

```
. clonevar airtrain = mode if mode == "air":travel | mode == "train":travel
```

For more details, see [U] [13.11 Label values](#).



Acknowledgments

`clonevar` was written by Nicholas J. Cox of the Department of Geography at Durham University, UK, who is coeditor of the *Stata Journal* and author of *Speaking Stata Graphics*. He in turn thanks Michael Blasnik of Nest Labs and Ken Higbee of StataCorp for very helpful comments on a precursor of this command.

Also see

[D] [generate](#) — Create or change contents of variable

[D] [separate](#) — Create separate variables

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.

